

9-сабақ — Стробоскоп, 1-бөлім: лента және түстер

«Стробоскоп» модулі · 45 минут · 3–8 сабақтар қажет

Танысыңыз: WS2812B

Машинкамыздың бірінші қақпаны — WS2812B адрестік жарық диодты лентасындағы стробоскоп. Оның қарапайым гирляндадан айырмашылығы неде? Гирляндада барлық шамдар — бір топ: хормен жанып, хормен сөнеді. Ал адрестік лентада ӘР жарық диодының өз кішкентай басқарушы чипі және өз нөмірі — адресі бар. Қаласаңыз — біріншісі қызыл жанады, үшіншісі көк, қалғандары үнсіз. Барлық керемет эффектілер осыдан жасалады.

Қосу

Лентаның үш сымы бар. Қуат — қызыл «+5V» және ақ/қара «—» — машинканың қуат таратқышының клеммаларына қосылады (лента көп жейді, оны ардуинодан емес, тікелей таратқыштан қоректендіреміз!). DIN сигналы (ортаңғы сым, жасыл) — ардуиноның D6 пиніне. Лентаның өзіндегі көрсеткілерге назар аударыңыз: сигнал тек бір бағытта жүгіреді, көрсеткілер «кіретін» жағынан қосыламыз.

[СУРЕТ ОРНЫ] Қосу сұлбасы: WS2812B лентасы, +5V және GND сымдары таратқыш клеммаларына, DIN сымы — ардуиноның D6 пиніне барады. Лентадағы бағыт көрсеткілері ерекшеленген.

Қуат өшірулі кезде қосамыз! Лента үшін «+» пен «—»-ті шатастыру — өлім. Көрсеткілерді тексеріңіз — «құйрығынан» лента жұмыс істемейді.

Ақыры #include құпиясын ашамыз

Лента чиптерін басқару — дәл микросекундтық импульстары бар күрделі іс. Бақытымызға орай, мұны біз үшін жасап қойған: бүкіл әлем бағдарламашылары кітапханалар — дайын функциялар жинақтарын жазады, оларды қолдануға болады. Лента үшін атақты FastLED («жылдам жарық диодтары») кітапханасын аламыз.

Функциялар кітабын өз бағдарламаңа қосу үшін #include <FastLED.h> жолын жазады. Оны мүшелеп талдайық. include сөзі — «қосу, енгізу». Басындағы решётка # — бұл команда платаға емес, бағдарлама ҚҰРАСТЫРУШЫСЫНА арналғанының белгісі: «менің кодымды құрастырмас бұрын, сөреден FastLED кітабын алып, бағдарламама жапсыр». Мұндай командалар платаға жүктелгенге дейін, компьютерде орындалады — сондықтан оларда ерекше белгі бар және нүктелі үтір жоқ. < > бұрыштық жақшалар «кітапты ортақ кітапхана сөресінен ізде» дегенді білдіреді, ал .h — «header», кітаптың мазмұны. Айтпақшы, бұл құрылымды көргенсіз: «Компилятор»

қойындысындағы шаблон `#include`-тен басталады — енді оның не екенін білесіз!

Массив — қорапшалардан тұратын шкаф

Бір айнымалы — бір қорапша. Ал сегіз жарық диодының түстерін бірден қалай сақтаймыз? Ол үшін массив бар — ортақ атпен нөмірленген қорапшалардың тұтас шкафы. `CRGB leds[KOL_LED];` жолы 8 қорапшалы `leds` шкафын жасайды, әрқайсысында — түс (`CRGB` типі «қызыл, жасыл және көктен тұратын түс» дегенді білдіреді). Қорапшаға нөмірмен қараймыз: `leds[0]` — біріншісі, `leds[7]` — соңғысы. Иә-иә, санақ нөлден — `for` циклындағыдай. Бәрі үйлеседі!

Тәжірибе: алғашқы жарық

```
#include <FastLED.h>

#define LENTA_PIN 6
#define KOL_LED 8

CRGB leds[KOL_LED];

void setup() {
    FastLED.addLeds<WS2812B, LENTA_PIN, GRB>(leds, KOL_LED);
    FastLED.setBrightness(60);
}

void loop() {
    leds[0] = CRGB::Red;
    FastLED.show();
}
```

`#define LENTA_PIN 6` жолы — «жапсырма-жариялау»: құрастырушы `LENTA_PIN` сөзін көрген жердің бәрінде 6 қояды. Ыңғайлы: сигнал сымы басқа пинге көшсе, бір цифрды өзгертеміз. `setup` ішіндегі `addLeds` командасы кітапхананы лентамызбен таныстырады (қай модель, қай пин, қанша жарық диоды), ал `setBrightness` жарықтықты шектейді — 255-тен 60-ы жетеді, толық жарықтық көзді қарықтырады және лентаны қыздырады. `loop` ішінде бірінші қорапшаға қызыл түс саламыз да, `FastLED.show()` командасымен бүкіл шкафты лентаға жібереміз. `show()`-сыз лента өзгермейді — түстер қорапшаларда жатыр, бірақ «пойыз әлі жіберілмеген»!

Енді бүкіл лентаны көкпен толтырамыз — `loop()` ішіндегісін ауыстырыңыз:

```
for (int i = 0; i < KOL_LED; i++) {
    leds[i] = CRGB::Blue;
}
FastLED.show();
```

Әр қорапшаға өз түсін беруге де болады (үш сан: қызыл, жасыл, көк, әрқайсысы 0-ден 255-ке дейін):

```
leds[0] = CRGB(255, 0, 0);  
leds[1] = CRGB(0, 255, 0);  
leds[2] = CRGB(255, 255, 0);  
leds[3] = CRGB(255, 120, 0);  
FastLED.show();
```

★ Тапсырмалар

1-тапсырма. Бүкіл лентаны өз фирмалық түсіңізбен жағыңыз — үш RGB санын таңдаңыз. CRGB(255, 255, 255) не береді? Ал CRGB(10, 10, 10)?

2-тапсырма. Жарық диодтарын «шахмат тәртібімен» жағыңыз: жұптары — жасыл, тақтары — өшірулі (CRGB::Black). Кеңес: циклде бірден аттап жүріңіз — $i = i + 2$.

3-тапсырма ★. Бағдаршам: бүкіл лента 3 секунд қызыл, бір секунд сары, 3 секунд жасыл, айналыммен. Түспен толтыруды функция етіп жазыңыз — келесі сабақта керек болады!